/**
 * Copyright (c) 2009 Teoti Graphix, LLC See:
 *     http://tkarchitecture.teotigraphix.org
 * 
 * Permission is hereby granted, free of charge, to any person
 * obtaining a copy of this software and associated documentation
 * files (the "Software"), to deal in the Software without
 * restriction, including without limitation the rights to use,
 * copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the
 * Software is furnished to do so, subject to the following
 * conditions:
 * 
 * The above copyright notice and this permission notice shall be
 * included in all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 * EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 * OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 * NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 * HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 * WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 * FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 * OTHER DEALINGS IN THE SOFTWARE.
 * 
 * @author Michael Schmalle mschmalle@teotigraphix.com
 */
package com.example.acme.view
{

import com.example.acme.ApplicationFacade;
import com.example.acme.model.proxy.UserProxy;
import com.example.acme.model.vo.User;
import com.example.acme.view.component.LoginView;

import flash.events.Event;

import org.puremvc.as3.multicore.interfaces.INotification;
import org.puremvc.as3.multicore.patterns.mediator.Mediator;

/**
 * The login component's mediator.
 * 
 * @author Michael Schmalle
 * @copyright Teoti Graphix, LLC
 */
public class LoginViewMediator extends Mediator
{
	//--------------------------------------------------------------------------
	//
	//  Public Class :: Properties
	//
	//--------------------------------------------------------------------------
	
	/**
	 * @private
	 */	
	public static const NAME:String = "LoginViewMediator";
	
	//--------------------------------------------------------------------------
	//
	//  Private :: Properties
	//
	//--------------------------------------------------------------------------
	
	/**
	 * The LoginView component.
	 */
	private function get loginView():LoginView
	{
		return viewComponent as LoginView;
	}	
	
	//--------------------------------------------------------------------------
	//
	//  Constructor
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Creates a new <code>LoginViewMediator</code> instance.
	 * 
	 * @param viewComponent The main LoginView application instance.
	 */
	public function LoginViewMediator( viewComponent:LoginView )
	{
		super( NAME, viewComponent );
	}
	
	//--------------------------------------------------------------------------
	//
	//  Overridden Public :: Methods
	//
	//--------------------------------------------------------------------------
	
	/**
	 * Register event listeners with the app and its fixed controls.
	 */
	override public function onRegister():void
	{
		loginView.addEventListener(LoginView.LOGIN, loginClick);
	}
	
	/**
	 * @private
	 */
	override public function listNotificationInterests():Array
	{
		return [
			ApplicationFacade.USER_LOGIN,
			ApplicationFacade.USER_LOGIN_COMPLETE,
			ApplicationFacade.USER_LOGIN_FAULT
		];
	}
	
	/**
	 * @private
	 */
	override public function handleNotification( note:INotification ):void
	{
		switch( note.getName() )
		{
			case ApplicationFacade.USER_LOGIN:
				var proxy:UserProxy = facade.retrieveProxy( 
					UserProxy.NAME ) as UserProxy;
				proxy.loginUser( note.getBody() as User );
				break;
			
			case ApplicationFacade.USER_LOGIN_COMPLETE:
				loginView.component.loginSucceeded();
				break;
			
			case ApplicationFacade.USER_LOGIN_FAULT:
				loginView.component.loginFailed();
				break;				
		}
	}
	
	/**
	 * @private
	 */
	private function loginClick(event:Event = null):void
	{
		var proxy:UserProxy = facade.retrieveProxy( 
			UserProxy.NAME ) as UserProxy;
		
		sendNotification( 
			ApplicationFacade.USER_LOGIN, 
			proxy.createTempUser( loginView.username, loginView.password ) );
	}		
}
}